
'---------------Title--------------

' File......16F877A_LCD2.pbp
' Started....3/18/06
' Microcontroller used:  Microchip Technology PIC16F877A
'                        microchip.com
' PicBasic Pro Code: micro-Engineering Labs, Inc.
'                    melabs.com 

'--------Program Desciption--------

' Demonstrates several commands to move LCD cursor
' on a 16 x 2 LCD diaplay.

'----------Related Lessons---------

' See LCD BASICS lesson at: 
' http://cornerstonerobotics.org/curriculum/lessons_year2/erii14_lcd1.pdf
'
' lcd2.pbp (the 16F88 program) is used in
' the lesson LCD Command Control Codes at:
' http://cornerstonerobotics.
org/curriculum/lessons_year2/erii15_lcd2_lcd_command_control_codes.pdf

'-----------Connections------------

' See schematic at:
' http://www.cornerstonerobotics.org/schematics/pic16f877a_lcd1_lcd2.pdf

'       16F877A Pin         Wiring
'       ---------           ---------- 
'       RA0                 LCD pin 11(DB4)
'       RA1                 LCD pin 12(DB5)
'       RA2                 LCD pin 13(DB6)
'       RA3                 LCD pin 14(DB7)
'       RA4                 LCD Register Select(RS)
'       RB3                 LCD Enable(E)
'       See schematic for the usual connections

'---------LCD Connections----------

'        LCD Pin              Wiring
'       ---------           ---------- 
'          1               Ground(Vss)
'          2               + 5v(Vdd)
'          3               Center of 20K Pot(Contrast)
'          4               RA4(Register Select,RS)
'          5               Ground(Read/Write,R/W)
'          6               RB3(Enable)
'          7               No Connection(DB0) 
'          8               No Connection(DB1)     
'          9               No Connection(DB2)     
'         10               No Connection(DB3)     
'         11               RA0(DB4)    
'         12               RA1(DB5)

'         13               RA2(DB6)
'         14               RA3(DB7)

'---------Revision History---------

' 11/28/07 Add 16F88 oscillator and ANSEL = 0
'          initializations
' 1/2/09   Change MCU from 16F88 to 16F877A
' 1/2/09   Delete ANSEL = 0 and add ADCON1 initialization

'-------------Variables------------

    c0 VAR BYTE                 ' Byte for counter
    c1 VAR BYTE                 ' Byte for second counter
         
'----------Initialization----------

    ADCON1 = %00000110          ' Changes PORTE and PORTA analog bits to 
                                ' digital operation since not using ADC
                                ' (Analog to Digital Converter)
                                        
' For the ADCON1 Register table, look at the 
' PIC16F877A datasheet.  For Microchip PIC datasheets:
' http://www.microchip.
com/stellent/idcplg?IdcService=SS_GET_PAGE&nodeId=2046 
' Select 8-bit PIC Microcontrollers, then the device from the
' drop down menu.  Now download the 16F87XA Datasheet.
' The ADCON1 Register is Register 11-2: ADCON1 Register,
' look around page 128 in the 16F877A datasheet.      

'-------------Main Code------------

    PAUSE 1000                  ' Pause to allow LCD to setup
    
start:          

    LCDOUT $FE,1,"Three"        ' Clears LCD screen, displays "Three"
    
    PAUSE 1000                  ' Pause 1 second 
    
    LCDOUT $fe,1                ' Clears LCD sreen
    
    LCDOUT $FE,$C0,"days"       ' Cursor moves to beginning of
                                ' second line, displays "days"       
    
    PAUSE 1000                  ' Pause 1 second 
    
    LCDOUT $fe,$14,"only"       ' Cursor moves to right one 
                                ' position and displays "only" 
    
    PAUSE 1000                  ' Pause 1 second       
    
    LCDOUT $fe,1                ' Clears LCD sreen      
    
    LCDOUT $fe,1,"Valid"        ' Clear LCD screen, display "Valid"

    PAUSE 1000                  ' Pause 1 second 
    
    FOR c0 = 1 TO 5             ' FOR..NEXT loop so "Valid" scrolls
                                ' off the screen to the left
    
    LCDOUT $fe,24               ' Scrolls display one character
                                ' position to the left
    
    PAUSE 100                   ' Pause 100 milliseconds
     
    NEXT                        ' Continue to next c0
    
    LCDOUT $fe,1,"in store only"' Clear LCD screen, display "in store only"
    
    PAUSE 1000                  ' Pause 1 second 
    
    FOR c0 = 1 TO 16            ' FOR..NEXT loop so "in store only"
                                ' scrolls off the screen to the right
    
    LCDOUT $fe,28               ' Scrolls display one character
                                ' position to the right
     
    PAUSE 100                   ' Pause 100 milliseconds
      
    NEXT                        ' Continue to next c0
         
    FOR c1 = 1 TO 2             ' FOR..NEXT loop to display
                                ' CD's  $5.00 two times
    
    LCDOUT $fe,1                ' Clear LCD screen
    
    LCDOUT $fe, $80+17,"CD's  $5.00"
                                ' Display "CD's  $5.00" starting
                                ' 17 positions from the beginning
                                ' of the first line, i.e., just off
                                ' LCD screen 
                                
    PAUSE 200                   ' Pause 1 second 
    
    FOR c0 = 1 TO 28            ' FOR..NEXT loop so "CD's  $5.00"
                                ' scrolls off the screen to the left
    
    LCDOUT $fe,24               ' Scrolls display one character
                                ' position to the left
    
    PAUSE 150                   ' Pause 150 milliseconds
     
    NEXT c0                     ' Continue to next c0 
    
    NEXT c1                     ' Continue to next c1 
    
    PAUSE 500                   ' Pause 500 milliseconds
    
    GOTO start                   ' Go to loop label 

    END

